
'CSE 1530 Winter 2006 Section M
'
' Solution to Exercise 5-5
' Wednesday, March 1 2006
'
' By: Bill Kapralos
'
' Note that the purpose of this program is to allow you to practise with
' strings and loops - it may not always produce gramaticaly correct results!

Option Explicit

' Definitions of any constants used throughout the code
Const C_UPPER As String = "C"
Const C_LOWER As String = "c"
Const FIND_STR As String = "ei"
Const REPLACE_STR As String = "ie"

' Exit button
Private Sub btnExit_Click()
   Unload Form1
   End
End Sub

' Clear button - reset the input and output textboxes
Private Sub btnClear_Click()
   txtInput.Text = vbNullString
   txtOutput.Text = vbNullString
End Sub

' "Check and Correct" button - perform all the work here
Private Sub btnCheck_Click()
   
   ' Local variable declarations
   Dim firstHalf As String
   Dim secondHalf As String
   Dim position As Integer
   Dim char As String
   Dim inputStr As String
   
   inputStr = txtInput.Text
   position = 1
   
   Do While (InStr(position, txtInput, FIND_STR) > 0)
      
      ' Obtain the position of the occurrence of "ei" (position of "e")
      position = InStr(position, inputStr, FIND_STR)
      
      ' Need to check for the case when "ei" are the first two characters of the
      ' input string.  If so, then ignore this since we cannot extract the
      ' previous character since it does not exist!
      If (position <> 1) Then
         
         ' Obtain the character one position before the occurrence of "ei"
         char = Mid(inputStr, position - 1, 1)
         
         ' Check if character is an uupper or lower case "c"
         If ((char <> C_LOWER) And (char <> C_UPPER)) Then

            ' Extract portion of string before the occurrence of "ei"
            ' e.g., from beginning of string to the occurrence of "c" before "ei"
            firstHalf = Mid(inputStr, 1, position - 1)
            
            ' Replace the "ei" with "ie" and extract everything to the left
            ' of "ie" including "ie" itself
            secondHalf = Replace(inputStr, FIND_STR, REPLACE_STR, position, 1)
            
            'catenate the two "halves"
            inputStr = firstHalf + secondHalf
         End If
      End If
      
      ' Check for the next occurrence of "ei" starting at the next position
      position = position + 1
   Loop
   
   'Display corrected string
   txtOutput.Text = inputStr
End Sub


